Computer algebra describes flow of turbulent floods via the Smagorinski large eddy closure

A. J. Roberts*

March 16, 2008

Abstract

Consider the turbulent flow of a layer of fluid. The Smagorinski closure for turbulence, with its linear dependence of eddy viscosity upon the shear-rate, models turbulent dissipation. A slow manifold model of the dynamics of the fluid layer allows for large changes in layer thickness provided the changes occur over a large enough lateral length scale. The slow manifold is based on two macroscopic modes by modifying the spectrum: here artificially modify the boundary conditions on the free surface so that, as well as a mode representing conservation of fluid, a lateral shear flow with slip is a neutral critical mode. Then remove the modification to recover a model for turbulent floods.

Contents

*Computational Engineering and Science Research Centre, Department of Mathematics & Computing, University of Southern Queensland, Toowoomba, Queensland 4352, Australia. mailto:aroberts@usq.edu.au

1 Introduction

This approach to modelling turbulent floods develops from that of modelling non-Newtonian fluids which have a nonlinear dependence upon strain rate (Roberts 2007*a*,*b*). Bijvelds et al. (1999) required enhanced lateral mixing which is naturally predicted by this slow manifold approach. Roberts et al. (2008) reports preliminary results using the model derived with the computer algebra program documented herein.

Consider the two-dimensional flow of a thin layer of turbulent fluid on a flat substrate. Let coordinate x measure distance along the substrate and coordinate z the distance normal to the substrate. Let the incompressible fluid have thickness $\eta(x,t)$, constant density ρ, and the nonlinear constitutive relation of the Smagorinski closure (Kim 2002, Marstop 2006, Özgökmen, Iliescu, Fischer, Srinivasan & Duan 2007, e.g.). The fluid flows with some varying velocity field $\mathbf{q} = (u, w)$ and pressure field p; these fields are the turbulent mean fields, that is, the fields averaged over realisations.

1.1 Uniform acceleration

Do not allow any lateral variations, $\partial_x = 0$. Then a low order slow manifold, errors $\mathcal{O}\big(\gamma^2 + g_x^2\big)$, is that in terms of the scaled vertical coordinate $\zeta = z/\eta$ the fluid fields are

$$w = 0\,, \quad \text{(shear flow)} \tag{1}$$

$$p = g_z(1-\zeta)\eta\,, \quad \text{(hydrostatic)} \tag{2}$$

$$\begin{aligned}
u &= \bar u\frac{2(\zeta+c_u)}{1+2c_u} \\
&\quad + \gamma\bar u\frac{(1+c_u)[(1+4c_u)(c_u+\zeta)-2(1+2c_u)(3c_u\zeta^2+\zeta^3)]}{4(1+2c_u)^2(1+3c_u+3c_u^2)} \\
&\quad + \frac{g_x\eta}{\bar u}\,\frac{[(5+6c_u)(c_u+\zeta)-6(2+7c_u+6c_u^2)\zeta^2+6(1+2c_u)^2\zeta^3]}{48\sqrt2\,c_t(1+3c_u+3c_u^2)}\,, \\
\dot\varepsilon &= \frac{\bar u}{\eta}\frac{\sqrt2}{1+2c_u} + \frac{\gamma\bar u}{\eta}\,\frac{\sqrt2(1+c_u)[(1+4c_u)-6(1+2c_u)(2c_u\zeta+\zeta^2)]}{8(1+2c_u)^2(1+3c_u+3c_u^2)} \\
&\quad + \frac{g_x}{\bar u}\,\frac{[(5+6c_u)-12(2+7c_u)\zeta+18(1+2c_u)^2\zeta^2]}{96c_t(1+3c_u+3c_u^2)}\,.
\end{aligned} \tag{3}$$

The parameter $c_u\eta$ is a 'slip' length on the bed, see boundary condition (13), and c_t parametrises the strength of Smagorinski's turbulent mixing, see the eddy viscosity (9); they are determined from observations. Figure 1 displays a sample of the vertical profile of the rate of strain $\dot\varepsilon$ and of the lateral



Figure 1: approximate vertical profiles at open channel flow equilibrium for which $g_x\eta/\bar u^2 = 0.0031$.

velocity u. This equilibrium flow resolves the shear in the lateral velocity and the increase in rate of strain near the bed. Our analysis does not attempt to resolve the turbulent log layer: we assume the details of dynamical interest are those determined by the relatively large scale of the fluid depth.

The set of such profiles in the vertical, and their nonlinear interactions, form a slow manifold. The evolution on this slow manifold of the mean lateral velocity $\bar u$ is dictated by turbulent bed drag limiting gravitational forcing:

$$\begin{aligned}\frac{d\bar u}{dt} =& -\frac{\sqrt{2}\,3c_t(1+c_u)}{(1+2c_u)(1+3c_u+3c_u^2)}\frac{\gamma\bar u^2}{\eta} + \frac{\frac{3}{4}+3c_u+3c_u^2}{1+3c_u+3c_u^2}g_x \\ &+ \mathcal{O}\left(\gamma^2+g_x^2+\partial_x\right) \end{aligned} \tag{4}$$

Upon putting the artificial parameter $\gamma = 1$ to recover the physical model, this evolution predicts an equilibrium channel flow at a mean velocity of

$$\bar{u} = \frac{1}{2}\left[\frac{(1+2c_u)^3}{\sqrt{2}c_t(1+c_u)}\right]^{1/2}\sqrt{g_x\eta}\,. \tag{5}$$

For example, choosing $c_t = 0.020$ and $c_u = 1.848 \approx 13/7 \approx 11/6$ gives about the correct channel flow *and* gives about the correct eddy viscosity when compared with observations of open channel flow (Nezu 2005, e.g.).

2 Overview

Denote free surface thickness $\eta(x,t)$ by h, mean lateral velocity $\bar{u}(x,t)$ by uu, and their evolution $\eta_t = $ gh and $\bar{u}_t = $ gu. The Reynolds number re, and the coefficients of lateral and normal gravitational forcing are Gravity numbers grx and grz. Construct an asymptotic solution of the Smagorinski equations in terms of η and $\bar{u}$ to some order of nonlinearity in $\bar{u}$ and some order of lateral derivatives ∂_x.

Decide upon how the asymptotic expansions of the solution are to be truncated. Then iteratively update the velocity and pressure fields to solve the Smagorinski equations and boundary conditions. The iteration continues until the governing equations are satisfied; that is, their residuals are zero to the order of truncation.

```
                    ⊳⊳ smag ⊲⊲
% see cadftf.pdf for documentation
on div; off allfac; on revpri; on gcd;
⊲⊲ preamble ⊳⊳
⊲⊲ initialise with linear ⊳⊳
⊲⊲ truncate the asymptotic expansion ⊳⊳
it:=1$
repeat begin ok:=1;
⊲⊲ solve continuity ⊳⊳
⊲⊲ nonlinear stress-strain relationships ⊳⊳
```

```
⊲⊲ solve vertical momentum and normal stress ⊳⊳
⊲⊲ solve horizontal momentum and FS stress ⊳⊳
⊲⊲ update thickness evolution ⊳⊳
showtime;
end until ok or (it:=it+1)>9;
number_iterations:=it;
⊲⊲ postprocess ⊳⊳
end;
```

3 Preamble

Improve printing by factoring with respect to these variables. It is a matter of taste and may be different depending upon what one wishes to investigate in the algebraic expressions.

⊳⊳ preamble ⊲⊲

```
factor uu,h,ct,gx,gz,gam,r2;
```

3.1 Define order parameters

Use the operator `h(m)` to denote m lateral derivatives of the fluid thickness, $\partial_x^m\eta$, and similarly `uu(m)` denotes m lateral derivatives of the mean shear, $\partial_x^m\bar{u}$. Also define readable abbreviations for η and its first spatial derivative. Note: use `d` to count the number of lateral x derivatives so we can easily truncate the asymptotic expansion.

⊳⊳ preamble ⊲⊲+

```
operator h; operator uu;
eta:=h(0); etax:=h(1)*d$
```

These operators must depend upon time and lateral space. Then lateral derivatives transform as ∂_x`h(m)` $=$ `h(m+1)` , for example. Also, a time derivative transforms into m lateral derivatives of the corresponding evolution: for example, ∂_t`h(m)` $= \partial_x^m$`gh` .

▷▷ preamble ◁◁+

```
depend h,xx,tt;
depend uu,xx,tt;
let { df(h(~m),xx) => h(m+1)
    , df(h(~m),xx,2) => h(m+2)
    , df(h(~m),tt) => df(gh,xx,m)
    , df(uu(~m),xx) => uu(m+1)
    , df(uu(~m),xx,2) => uu(m+2)
    , df(uu(~m),tt) => df(gu,xx,m) };
```

3.2 Stretch the coordinates with the free surface

Use stretched coordinates `zz`, `xx` and `tt` to denote $Z = z/\eta(x,t)$, $X = x$ and $T = t$. The free surface is then simply $Z = 1$.

▷▷ preamble ◁◁+

```
depend xx,x,z,t;
depend zz,x,z,t;
depend tt,x,z,t;
```

Then space-time derivatives transform according to

$$\frac{\partial}{\partial x} = \frac{\partial}{\partial X} - Z\frac{\eta_X}{\eta}\frac{\partial}{\partial Z}, \quad \frac{\partial}{\partial t} = \frac{\partial}{\partial T} - Z\frac{\eta_t}{\eta}\frac{\partial}{\partial Z}, \quad \frac{\partial}{\partial z} = \frac{1}{\eta}\frac{\partial}{\partial Z}.$$

I neatly insert an automatic count of lateral x derivatives here, with `d`, in between ∂_x and ∂_X.

```
▷▷ preamble ◁◁+
let { df(~a,x) => df(a,xx)*d-zz*etax/eta*df(a,zz)
    , df(~a,t) => df(a,tt)-zz*gh/eta*df(a,zz)
    , df(~a,z) => df(a,zz)/eta
    , df(~a,x,2) => df(df(a,x),x) };
```

4 Initialise with linear

Start the iteration from the linear solution that the lateral velocity $u = \bar{u}(c_u+\zeta)/(c_u+\frac{1}{2})$ and all other fields are zero, $w = p = 0$. The parameter c_u determines the turbulent slip on the bed and is to be determined to best fit experiment and/or observations. The evolution of the 'order parameters' is also zero: $\bar{u}_t = \mathtt{gu} = 0$ and $\eta_t = \mathtt{gh} = 0$.

```
▷▷ initialise with linear ◁◁
let r2^2=>2; % r2=sqrt2
u:=uu(0)*(cu+zz)/(cu+1/2);
w:=p:=gh:=gu:=0;
```

5 Truncate the asymptotic expansion

There are lots of ways to truncate the asymptotic model. The small parameters available are:

- `d` counting the number of x derivatives of the slowly varying lateral spatial structure in any term;
- the homotopy parameter `gam` varying between $\gamma = 0$ for the artificial base problem and $\gamma = 1$ for the physical fluid equations; and
- `grx` and `grz` being the lateral and normal components of gravity.

Note: the velocity of the flow is not small; consider the parameter $\bar{u}$ finite.

Usually we will make lateral gravity fairly small by scaling with the magnitude of ∂_x.

Initially omit all x derivatives by setting $d = 0$, later we scale d with `eps` to get the relatively simple but interesting model with errors $\mathcal{O}\big(\gamma^{3/2} + g_x^{3/2} + g_z^3 + \partial_x^3\big)$.[1] We need not make normal gravity g_z small as here, but doing so removes some messy terms.

```
▷▷ truncate the asymptotic expansion ◁◁
d:=eps;
grz:=eps*gz;
grx:=eps^2*gx;
gamm:=eps^2*gam;
factor eps;
```

For now truncate to relatively low order, $\mathcal{O}\big(\gamma^{3/2} + g_x^{3/2} + g_z^3 + \partial_x^3\big)$, in spatial derivatives and boundary condition artifice (can do $\mathcal{O}\big(\epsilon^4\big)$ if no spatial variations):

```
▷▷ truncate the asymptotic expansion ◁◁+
let { eps^3=>0 };
```

6 Update w with continuity and no flow through bed

The nondimensional PDEs for the incompressible fluid flow include the continuity equation

$$\nabla \cdot \mathbf{q} = \frac{\partial u}{\partial x} + \frac{\partial w}{\partial z} = 0\,, \tag{6}$$

[1] We need to check the algorithm at $\mathcal{O}\big(g_x^3\big)$.

to be solved with no flow through the bed,

$$w = 0 \quad \text{on} \quad z = 0\,. \tag{7}$$

As with all field variables in this model, the quantities u, w and p are averaged over the ensemble of turbulent flows. Compute the residual of the continuity equation, then update the vertical velocity w by integrating from the bed, $\zeta = 0$. The variable `ok` stores whether all residuals are so far zero in this iteration.

▷▷ solve continuity ◁◁

```
resc:=df(u,x)+df(w,z);
ok:=if ok and resc=0 then 1 else 0;
w:=w-eta*wsolv(resc,zz);
```

Use the linear operator `wsolv` as it is quicker than native integration.

▷▷ preamble ◁◁+

```
operator wsolv; linear wsolv;
let {wsolv(zz^~n,zz) => zz^(n+1)/(n+1)
    ,wsolv(zz,zz) => zz^2/2
    ,wsolv(1,zz) => zz };
```

7 Smagorinski large eddy stress-shear closure

Now the strain-rate tensor

$$\dot\varepsilon_{ij} = \frac{1}{2}\left(\frac{\partial u_i}{\partial x_j} + \frac{\partial u_j}{\partial x_i}\right)\,.$$

```
⊳⊳ nonlinear stress-strain relationships ⊲⊲
exx:=df(u,x);
ezz:=df(w,z);
exz:=(df(u,z)+df(w,x))/2;
```

Then the stress tensor for the fluid is $\sigma_{ij} = -p\delta_{ij} + 2\rho\nu\dot\varepsilon_{ij}$: when the viscosity ν is constant this models a Newtonian fluid; but here the Smagorinski closure for turbulent flow is that this eddy viscosity varies linearly with strain-rate magnitude (analogous to a shear thickening non-Newtonian fluid). Define and compute the magnitude `ros` $= |\dot\varepsilon|$, the second invariant of the strain-rate tensor, where

$$|\dot\varepsilon|^2 = \sum_{i,j} \dot\varepsilon_{ij}^2\,. \tag{8}$$

```
⊳⊳ nonlinear stress-strain relationships ⊲⊲+
rese:=exx^2+2*exz^2+ezz^2-ros^2;
ok:=if ok and rese=0 then 1 else 0;
ros:=ros+rese*eta*(cu+1/2)/r2/uu(0);
```

Initially approximate the magnitude $\dot\varepsilon$ of the strain-rate tensor: the above iteration step assumes the strain rate is $\bar u\sqrt{2}/\eta/(1+2c_u)$ to leading approximation.

```
⊳⊳ initialise with linear ⊲⊲+
ros:=uu(0)*r2/eta/(1+2*cu);
```

Approximate the eddy viscosity at any point in the fluid as proportional to the local strain-rate magnitude,

$$\nu = c_t\eta^2\dot\varepsilon\,, \tag{9}$$

where c_t is a dimensionless constant to be chosen to fit experiments and/or observations. In the Smagorinski model (Özgökmen, Iliescu, Fischer, Srinivasan & Duan 2007, e.g.) $c_t = (c_s\Delta/\eta)^2$ where arguments indicate $c_s \approx 0.2$.

To match observations (Nezu 2005, e.g.) of open channel flow equilibria we set $c_t = 0.020$ from which the appropriate filter scale $\Delta \approx 0.7\eta$ consistent with significant mixing across the fluid layer as seen in Figures 14–15 by Janosi et al. (2004).

```
              ▷▷ initialise with linear ◁◁+
ct:=1/50;
```

For whatever c_t is chosen, the deviatoric stress tensor is $\tau_{ij} = 2\nu\dot\varepsilon_{ij} = 2c_t\eta^2\dot\varepsilon\dot\varepsilon_{ij}$.

```
       ▷▷ nonlinear stress-strain relationships ◁◁+
txx:=2*ct*eta^2*ros*exx;
tzz:=2*ct*eta^2*ros*ezz;
txz:=2*ct*eta^2*ros*exz;
```

8 Update pressure from vertical momentum and surface normal stress

The nondimensional momentum equation is

$$\frac{\partial \mathbf{q}}{\partial t} + \mathbf{q}\cdot\nabla\mathbf{q} = -\nabla p + \nabla\cdot\boldsymbol{\tau} + \mathbf{g}\,, \tag{10}$$

where $\boldsymbol{\tau}$ is the nondimensional deviatoric eddy stress tensor, and $\mathbf{g}$ is the direction of gravity; when the substrate slopes, $\mathbf{g}$ is not normal to the substrate. The vertical momentum equation is solved with the surface condition that the turbulent mean, normal stress to the free surface is zero, that is,

$$-p + \frac{1}{1+\eta_x^2}\left(\tau_{zz} - 2\eta_x\tau_{xz} + \eta_x^2\tau_{xx}\right) = 0 \quad \text{on} \quad z = \eta\,. \tag{11}$$

Compute the residuals of the vertical momentum equation and the zero normal stress on the free surface.

▷▷ solve vertical momentum and normal stress ◁◁

```
resw:=df(w,t)+u*df(w,x)+w*df(w,z)
    +df(p,z) +grz -df(txz,x)-df(tzz,z);
restn:= sub(zz=1,-p*(1+etax^2) +tzz
    -2*etax*txz +etax^2*txx );
ok:=if ok and {resw,restn}={0,0} then 1 else 0;
```

Update the pressure field p by integrating down from the free surface $\zeta = 1$; we use the linear operator `psolv` to solve $\partial_\zeta p' = -\text{RHS}$ such that $p' = 0$ at $\zeta = 1$.

▷▷ solve vertical momentum and normal stress ◁◁+

```
p:=p+eta*psolv(resw,zz)+restn;
```

It is quicker to use operators than to use the native integration.

▷▷ preamble ◁◁+

```
operator psolv; linear psolv;
let {psolv(zz^~n,zz) => (1-zz^(n+1))/(n+1)
    ,psolv(zz,zz) => (1-zz^2)/2
    ,psolv(1,zz) => (1-zz) };
```

9 Update u from horizontal momentum and surface tangential stress

There must be no turbulent mean, tangential stress at the free surface,

$$(1-\eta_x^2)\tau_{xz} + \eta_x(\tau_{zz} - \tau_{xx}) = 0 \quad \text{on} \quad z = \eta\,. \tag{12}$$

Also, put a slip law on the mean bed to provide bed drag:

$$u = c_u \eta \frac{\partial u}{\partial z} \quad \text{on } z = 0\,, \tag{13}$$

for some constant $c_u \approx 11/6$ to match open channel flow observations.

▷▷ initialise with linear ◁◁+

```
cu:=11/6;
```

To get centre manifold theory support for the slow manifold model of shallow water flow, modify the surface condition (12) on the tangential stress to have an artificial forcing proportional to the square of the local, free surface, velocity:

$$\left[(1-\eta_x^2)\tau_{xz}+\eta_x(\tau_{zz}-\tau_{xx})\right] = \frac{(1-\gamma)c_t}{\sqrt{2}(1+c_u)^2}u^2 \quad \text{on } z=\eta\,. \tag{14}$$

When we evaluate at $\gamma=1$ this artificial right-hand side becomes zero so the artificial surface condition (14) reduces to the physical surface condition (12). However, when both the parameter $\gamma=0$ and the lateral derivatives are negligible, $\partial_x=0$, then the lateral shear $u \propto c_u+\zeta$ becomes a neutral mode of the dynamics.

The Euler parameter of a toy problem suggests introducing a factor $(1-\frac{1}{6}\gamma)$ into the left-hand side of the tangential stress boundary condition (14) in order to improve convergence in the parameter γ when evaluated at the physically relevant $\gamma=1$. This needs further exploration. For the moment omit such a factor.

Compute the residuals of the lateral momentum equation, an artificial tangential stress on the free surface, and the bed boundary. See that when $\gamma=0$ the free surface condition is effectively $\eta\partial_z u = u/(1+c_u)$, leading to our neutral mode $u \propto c_u+\zeta$, namely $u=\bar u(x,t)(c_u+\zeta)/(c_u+\frac{1}{2})$; whereas when $\gamma=1$ the free surface condition reduces to zero tangential stress.

▷▷ solve horizontal momentum and FS stress ◁◁

```
resu:= df(u,t)+u*df(u,x)+w*df(u,z)
    +df(p,x) -grx -df(txx,x)-df(txz,z);
restt:=-sub(zz=1,
    (1-0*gamm)*((1-etax^2)*txz+etax*(tzz-txx))
```

```
    -(1-gamm)*r2*ct/(cu+1)^2/2*u^2);
resb:=sub(zz=0,-u+cu*eta*df(u,z));
ok:=if ok and {resu,restt,resb}={0,0,0} then 1 else 0;
```

Use these residuals to update the lateral velocity field u and the evolution of the mean shear $\bar{u}$. First update the evolution with some magic recipe depending upon the boundary residuals and the mean of the residual of the u equation. Second update the lateral velocity field using operator `usolv`.

```
▷▷ solve horizontal momentum and FS stress ◁◁+
gu:=gu+(gud:=-ca*mean(resu*(cu+zz),zz)+cc*restt/eta );
u:=u+usolv(resu+(cu+zz)/(cu+1/2)*gud,zz)*cb*eta/uu(0);
```

Using these coefficients.

```
▷▷ preamble ◁◁+
ca:=3*(cu+1/2)/(1+3*cu+3*cu^2);
cc:=3*(cu+1/2)*(cu+1)/(1+3*cu+3*cu^2);
cb:=(cu+1/2)/(r2*ct);
```

The linear operator `mean` quickly computes the average of some field over the fluid thickness.

```
▷▷ preamble ◁◁+
operator mean; linear mean;
let { mean(zz^~n,zz) => 1/(n+1)
    , mean(zz,zz) => 1/2
    , mean(1,zz) => 1 };
```

The linear operator `usolv` solves $\partial_\zeta^2 u' = \text{RHS}$ such that the bed boundary condition (13) is always satisfied and that the mean of the solution u' is always zero to ensure $\bar{u}$ remains the mean later velocity.

⊳⊳ preamble ⊲⊲+

```
operator usolv; linear usolv;
let { usolv(zz^~n,zz) => (zz^(n+2)
          -(cu+zz)/(n+3)/(cu+1/2) )/(n+2)/(n+1)
    , usolv(zz,zz) => (zz^3 -(cu+zz)/4/(cu+1/2) )/6
    , usolv(1,zz) => (zz^2 -(cu+zz)/3/(cu+1/2) )/2 };
```

10 Update the free surface evolution

The kinematic condition at the free surface,

$$\frac{\partial \eta}{\partial t} + u\frac{\partial \eta}{\partial x} = w \quad \text{on} \quad z = \eta\,, \tag{15}$$

gives the evolution of the fluid thickness h.

⊳⊳ update thickness evolution ⊲⊲

```
gh:=sub(zz=1,w-u*etax);
```

11 Postprocessing

I may use these transformations to check on the dimensionality of various expressions. But not at the moment.

⊳⊳ postprocess ⊲⊲

```
dims:={ h(~m)=>nh*ll
    , uu(~m)=>mu*ll/tt
    , gx=>ngx*ll/tt^2 }$
```

Write out the final evolution on the slow manifold.

```
              ▷▷ postprocess ◁◁+
r2:=sqrt(2)$ eps:=1$
on rounded; print_precision 4;
write dhdt:=gh;
write dudt:=gu;
```

12 Trace the execution

I like to see how the iteration is proceeding. For each equation, write out the number of terms in its residual throughout iteration.

```
               ▷▷ preamble ◁◁+
procedure mylength(res); %res;
if res=0 then 0 else length(res);
```

```
           ▷▷ solve continuity ◁◁+
write resc:=mylength(resc);
```

```
  ▷▷ nonlinear stress-strain relationships ◁◁+
write rese:=mylength(rese);
```

```
 ▷▷ solve vertical momentum and normal stress ◁◁+
write resw:=mylength(resw);
write restn:=mylength(restn);
```

```
  ▷▷ solve horizontal momentum and FS stress ◁◁+
write resu:=mylength(resu);
```

```
write restt:=mylength(restt);
write resb:=mylength(resb);
```

References

Bijvelds, M. D. J. P., Kranenburg, C. & Stelling, G. S. (1999), '3d numerical simulation of turbulent shallow-water in square harbor', *Journal of Hydraulic Engineering* **125**(1), 26–31. 2

Janosi, I. M., Jan, D., Szabo, K. G. & Tel, T. (2004), 'Turbulent drag reduction in dam-break flows', *Experiments in Fluids* **37**, 219–229. 12

Kim, T. K. (2002), A Modified Smagorinsky Subgrid Scale Model for the Large Eddy Simulation of Turbulent Flow, PhD thesis, Univeristy of California, Davis. [http://arXiv.org/abs/math.sc/0209377]. 3

Marstop, L. (2006), Subgrid-scale modelling for large-eddy simulation including scalar mixing in rotating turbulent shear flows, PhD thesis, Royal Institute of Technology KTH Mechanics SE-100 44 Stockholm, Sweden. 3

Nezu, I. (2005), 'Open-channel flow turbulence and its research prospect in the 21st century', *J. Hydraulic Engineering* p. 229. doi:10.1061/(ASCE)0733-9429(2005)131:4(229). 5, 12
ΩÖzgökmen et al.

Özgökmen, T. M., Iliescu, T., Fischer, P. F., Srinivasan, A. & Duan, J. (2007), 'Large eddy simulation of stratified mixing and two-dimensional dam-break problem in a rectagular enclosed domain', *Ocean Modelling* **16**, 106–140. 3, 11

Roberts, A. J. (2007*a*), Computer algebra models the inertial dynamics of a thin film flow of power law fluids and other generalised newtonian fluids, Technical report, University of Southern Queensland. http://eprints.

usq.edu.au/archive/00002010/.
URL: *http://eprints.usq.edu.au/archive/00002010/* 2

Roberts, A. J. (2007*b*), 'The inertial dynamics of thin film flow of non-newtonian fluids', *Physics Letters A* **372**, 1607–1611. 2

Roberts, A. J., Georgiev, D. J. & Strunin, D. V. (2008), Model turbulent floods with the Smagorinski large eddy closure, Technical report, University of Southern Queensland. 2